

NAG Library Routine Document

F08FSF (ZHETRD)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F08FSF (ZHETRD) reduces a complex Hermitian matrix to tridiagonal form.

2 Specification

```
SUBROUTINE F08FSF (UPLO, N, A, LDA, D, E, TAU, WORK, LWORK, INFO)
INTEGER                N, LDA, LWORK, INFO
REAL (KIND=nag_wp)    D(*), E(*)
COMPLEX (KIND=nag_wp) A(LDA,*), TAU(*), WORK(max(1,LWORK))
CHARACTER(1)          UPLO
```

The routine may be called by its LAPACK name *zhetrd*.

3 Description

F08FSF (ZHETRD) reduces a complex Hermitian matrix A to real symmetric tridiagonal form T by a unitary similarity transformation: $A = QTQ^H$.

The matrix Q is not formed explicitly but is represented as a product of $n - 1$ elementary reflectors (see the F08 Chapter Introduction for details). Routines are provided to work with Q in this representation (see Section 9).

4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Arguments

- 1: UPLO – CHARACTER(1) *Input*
On entry: indicates whether the upper or lower triangular part of A is stored.
UPLO = 'U'
The upper triangular part of A is stored.
UPLO = 'L'
The lower triangular part of A is stored.
Constraint: UPLO = 'U' or 'L'.
- 2: N – INTEGER *Input*
On entry: n , the order of the matrix A .
Constraint: $N \geq 0$.
- 3: A(LDA,*) – COMPLEX (KIND=nag_wp) array *Input/Output*
Note: the second dimension of the array A must be at least $\max(1, N)$.
On entry: the n by n Hermitian matrix A .

If UPLO = 'U', the upper triangular part of A must be stored and the elements of the array below the diagonal are not referenced.

If UPLO = 'L', the lower triangular part of A must be stored and the elements of the array above the diagonal are not referenced.

On exit: A is overwritten by the tridiagonal matrix T and details of the unitary matrix Q as specified by UPLO.

- 4: LDA – INTEGER *Input*
On entry: the first dimension of the array A as declared in the (sub)program from which F08FSF (ZHETRD) is called.
Constraint: $LDA \geq \max(1, N)$.
- 5: D(*) – REAL (KIND=nag_wp) array *Output*
Note: the dimension of the array D must be at least $\max(1, N)$.
On exit: the diagonal elements of the tridiagonal matrix T .
- 6: E(*) – REAL (KIND=nag_wp) array *Output*
Note: the dimension of the array E must be at least $\max(1, N - 1)$.
On exit: the off-diagonal elements of the tridiagonal matrix T .
- 7: TAU(*) – COMPLEX (KIND=nag_wp) array *Output*
Note: the dimension of the array TAU must be at least $\max(1, N - 1)$.
On exit: further details of the unitary matrix Q .
- 8: WORK(max(1, LWORK)) – COMPLEX (KIND=nag_wp) array *Workspace*
On exit: if INFO = 0, the real part of WORK(1) contains the minimum value of LWORK required for optimal performance.
- 9: LWORK – INTEGER *Input*
On entry: the dimension of the array WORK as declared in the (sub)program from which F08FSF (ZHETRD) is called.
 If LWORK = -1, a workspace query is assumed; the routine only calculates the optimal size of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.
Suggested value: for optimal performance, $LWORK \geq N \times nb$, where nb is the optimal **block size**.
Constraint: $LWORK \geq 1$ or $LWORK = -1$.
- 10: INFO – INTEGER *Output*
On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed tridiagonal matrix T is exactly similar to a nearby matrix $(A + E)$, where

$$\|E\|_2 \leq c(n)\epsilon\|A\|_2,$$

$c(n)$ is a modestly increasing function of n , and ϵ is the *machine precision*.

The elements of T themselves may be sensitive to small perturbations in A or to rounding errors in the computation, but this does not affect the stability of the eigenvalues and eigenvectors.

8 Parallelism and Performance

F08FSF (ZHETRD) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F08FSF (ZHETRD) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of real floating-point operations is approximately $\frac{16}{3}n^3$.

To form the unitary matrix Q F08FSF (ZHETRD) may be followed by a call to F08FTF (ZUNGTR):

```
CALL ZUNGTR(UPLO,N,A,LDA,TAU,WORK,LWORK,INFO)
```

To apply Q to an n by p complex matrix C F08FSF (ZHETRD) may be followed by a call to F08FUF (ZUNMTR). For example,

```
CALL ZUNMTR('Left',UPLO,'No Transpose',N,P,A,LDA,TAU,C,LDC, &
           WORK,LWORK,INFO)
```

forms the matrix product QC .

The real analogue of this routine is F08FEF (DSYTRD).

10 Example

This example reduces the matrix A to tridiagonal form, where

$$A = \begin{pmatrix} -2.28 + 0.00i & 1.78 - 2.03i & 2.26 + 0.10i & -0.12 + 2.53i \\ 1.78 + 2.03i & -1.12 + 0.00i & 0.01 + 0.43i & -1.07 + 0.86i \\ 2.26 - 0.10i & 0.01 - 0.43i & -0.37 + 0.00i & 2.31 - 0.92i \\ -0.12 - 2.53i & -1.07 - 0.86i & 2.31 + 0.92i & -0.73 + 0.00i \end{pmatrix}.$$

10.1 Program Text

```
Program f08fsfe
```

```
!      F08FSF Example Program Text
!
!      Mark 26 Release. NAG Copyright 2016.
!
!      .. Use Statements ..
!      Use nag_library, Only: nag_wp, zhetrd
!      .. Implicit None Statement ..
!      Implicit None
!      .. Parameters ..
!      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
!      Integer                     :: i, info, lda, lwork, n
```

```

Character (1)                                :: uplo
! .. Local Arrays ..
Complex (Kind=nag_wp), Allocatable :: a(:,,:), tau(:), work(:)
Real (Kind=nag_wp), Allocatable :: d(:), e(:)
! .. Intrinsic Procedures ..
Intrinsic                                     :: abs
! .. Executable Statements ..
Write (nout,*) 'F08FSF Example Program Results'
! Skip heading in data file
Read (nin,*)
Read (nin,*) n
lda = n
lwork = 64*n
Allocate (a(lda,n),tau(n-1),work(lwork),d(n),e(n-1))

! Read A from data file and copy A into C

Read (nin,*) uplo
If (uplo=='U') Then
  Read (nin,*)(a(i,i:n),i=1,n)
Else If (uplo=='L') Then
  Read (nin,*)(a(i,1:i),i=1,n)
End If

! Reduce A to tridiagonal form
! The NAG name equivalent of zhetrd is f08fsf
Call zhetrd(uplo,n,a,lda,d,e,tau,work,lwork,info)

If (info==0) Then
! Print the diagonal and off-diagonal of tridiagonal T.
! The absolute value of E is printed since this can vary by a change of
! sign (corresponding to multiplying through a column of Q by -1).

  Write (nout,*)
  Write (nout,*)
  'Diagonal and off-diagonal elements of tridiagonal form'
  Write (nout,*)
  Write (nout,99999) 'i', 'D', 'E'
  Do i = 1, n - 1
    Write (nout,99998) i, d(i), abs(e(i))
  End Do
  Write (nout,99998) n, d(n)

Else
  Write (nout,99997) info
End If

99999 Format (5X,A,9X,A,12X,A)
99998 Format (1X,I5,2(1X,F12.5))
99997 Format (1X,'** ZHETRD/F08FSF retuned with INFO = ',I10)

End Program f08fsfe

```

10.2 Program Data

F08FSF Example Program Data

```

4                                     :Value of N
'L'                                   :Value of UPLO
(-2.28, 0.00)
( 1.78, 2.03) (-1.12, 0.00)
( 2.26,-0.10) ( 0.01,-0.43) (-0.37, 0.00)
(-0.12,-2.53) (-1.07,-0.86) ( 2.31, 0.92) (-0.73, 0.00) :End of matrix A

```

10.3 Program Results

F08FSF Example Program Results

Diagonal and off-diagonal elements of tridiagonal form

i	D	E
1	-2.28000	4.33846
2	-0.12846	2.02259
3	-0.16659	1.80232
4	-1.92495	
